

## Università degli Studi di Verona

ARCHITETTURA DEGLI ELABORATORI

# Elaborato Assembly

Mirko Morati, Noè Murr

## Indice

1	Premessa	2
2	Descrizione del progetto	2
3	syscall.inc	2
4	main.s         4.1 Flowchart          4.2 Variabili Globali          4.3 Variabili Locali          4.4 Funzioni ed Etichette	4
5	open_files.s5.1 Variabili Locali5.2 Funzioni ed Etichette	
6	read_line.s6.1 Variabili Locali6.2 Funzioni ed Etichette	<b>5</b> 5
7	atoi.s 7.1 Funzioni ed Etichette	<b>6</b>
8	check.s 8.1 Funzioni ed Etichette	<b>6</b>
9	write_line.s 9.1 Variabili Locali	
10	itoa.s 10.1 Funzioni ed Etichette	<b>8</b>
11	close_files.s	8
12	12.4 read_line.s	9 9 11 12 14 15
	12.8 itoa.s	17 19 20

### 1 Premessa

Per la corretta lettura del pdf si informa che sono stati inseriti dei riferimenti per ogni etichetta e variabile trattata in modo da poter vedere e approfondire il corrispettivo codice nel programma.

## 2 Descrizione del progetto

Si vuole realizzare un programma Assembly per il monitoraggio di un motore a combustione interna il quale, ricevuto come ingresso il numero di giri/minuto del motore, fornisca in uscita la modalità di funzionamento corrente del motore: Sotto Giri, Ottimale, Fuori Giri. Il programma deve contare e visualizzare in uscita il numero dei secondi trascorsi nella modalità di funzionamento attuale ed inoltre attivare il segnale di allarme nel caso in cui il motore si trovi in modalità Fuori Giri da più di 15 secondi.

Di seguito verranno descritte le funzioni presenti in ogni file del programma, etichette, eventuali variabili e loro scopo.

## 3 syscall.inc

Header file contenente la definizione di alcune costanti, tramite la pseudo-operazione .equ, relative alle chiamate di sistema e ad alcuni standard utilizzati in tutti i file e riportati di seguito:

SYS_EXIT	1
SYS_READ	3
SYS_WRITE	4
SYS_OPEN	5
SYS_CLOSE	6
STDIN	0
STDOUT	1
STDERR	2
SYSCALL	0x80


## Scelte Progettuali

Nonostante il codice Assembly sia fortemente legato alla macchina sottostante, si è deciso di utilizzare un "trucchetto" per permetterne, per quanto possibile, la portabilità agevole su un'altra piattaforma. Infatti sarà sufficiente cambiare i codici delle SYS\_CALL nel file syscall.inc per utilizzare le SYS\_CALL di un altro sistema operativo. Questo vale solo su sistemi unix-like poiché i parametri delle funzioni essenziali (read, write, ...) sono comuni grazie allo standard POSIX.

## 4 main.s

File principale del programma.

## 4.1 Flowchart



### 4.2 Variabili Globali

- · input\_fd: Contiene il descrittore del file di input;
- · output\_fd: Contiene il descrittore del file di output;
- · init: Contiene il valore del segnale INIT corrente;
- · reset: Contiene il valore del segnale RESET corrente;
- · rpm: Contiene il valore del segnale RPM corrente;
- · alm: Contiene il valore del segnale ALM corrente;
- · mod: Contiene il valore del segnale MOD corrente;
- · numb: Contiene il valore del segnale NUMB corrente.

## Scelte Progettuali

Si è scelto di utilizzare in modo estensivo le variabili globali in quanto in un progetto di così ridotte dimensioni si aumenta notevolmente la leggibilità del codice e nel contempo se ne riduce la complessità di scrittura. Se il progetto avesse richiesto computazioni più complesse questo approccio non sarebbe stato l'ideale, poiché sarebbero state compromesse le prestazioni totali dell'applicazione.

#### 4.3 Variabili Locali

- · usage: Stringa per la descrizione del corretto utilizzo del programma;
- · USAGE\_LENGTH: Contiene la lunghezza della stringa, necessaria per la stampa.

#### 4.4 Funzioni ed Etichette

- · \_start: Punto di entrata del programma. Si occupa di controllare che il numero di parametri sia corretto, in caso contrario stampa la stringa usage e termina. Dopo il controllo chiama la funzione \_open\_files definita nel file open\_files.s.
- · \_main\_loop: Loop principale. Viene chiamata la funzione \_read\_line definita nel file read\_line.s, nel caso in cui il contenuto del registro ebx sia equivalente a -1 significa che il file di input è terminato (EOF → End Of File) quindi salta a \_end, altrimenti chiama la funzione \_check definita nel file check.s e la funzione \_write\_line definita nel file write\_line.s, dopodiché riesegue il ciclo.
- · \_end: Si occupa di chiudere tutti i file aperti e della corretta uscita dal programma tramite la chiamata di sistema exit.
- · \_show\_usage: Nel caso in cui i parametri non siano corretti stampa a video la stringa usage e termina il programma segnalando errore con il codice 1.

## 5 open files.s

Contiene la funzione che si occupa di aprire i file in modo corretto.

#### 5.1 Variabili Locali

- · error\_opening\_files: Stringa di errore in caso di errata apertura dei file (e.g. file mancante, file corrotto, ...).
- · ERROR\_OPENING\_LENGTH: Costante che contiene la lunghezza della stringa di errore.

#### 5.2 Funzioni ed Etichette

- · \_open\_files: Si occupa di aprire i file e gestisce eventuali errori. In caso il file di output non esistesse, questo viene creato in automatico con permessi di lettura e scrittura. I descrittori ottenuti vengono salvati nelle corrispondenti variabili globali.
- · \_error\_opening\_files: In caso di errore viene stampato su STDERR la stringa opportuna, dopodiché il programma viene terminato con codice di errore 2.

## 6 read line.s

Contiene la funzione che si occupa di leggere ed interpretare una riga per volta del file di input.

#### 6.1 Variabili Locali

- · input\_buff: Buffer di dimensione INPUT\_BUFF\_LEN che conterrà i caratteri della riga letta dal file di input.
- $\cdot$  INPUT\_BUFF\_LEN: Dimensione del buffer.

### Scelte Progettuali

Si è scelto di utilizzare un buffer di dimensione 9 byte corrispondente alla lunghezza di una singola riga. Questa scelta è molto importante perché permette di non realizzare soluzioni hardcoded. La consegna specifica che non verranno utilizzati file di input con più di 100 righe: sfruttando tale informazione si sarebbe potuto scrivere un codice semplicistico che leggesse in un buffer l'intero file. Soluzione poco elegante e mal ottimizzata. Un'altra strada prendibile sarebbe stata quella di leggere un byte alla volta dal file di input, ma questa avrebbe richiesto un numero spropositato e inutile di chiamate al kernel. Con la lettura di una riga per volta il buffer rimane snello e comodo da manipolare. Questa soluzione permette all'applicazione di lavorare con file di dimensione arbitraria, inoltre lo riteniamo un metodo ottimale in quanto stabilisce un compromesso tra l'uso di SYS\_CALL e flessibilità del codice. Oltre a questo si sarebbe potuto utilizzare un ulteriore metodo: quello di mappare in memoria il file ed accedervi come in un array, ma questo sforava dalle nozioni fornite dal corso.

### 6.2 Funzioni ed Etichette

- · \_read\_line: Legge dal file una riga tramite la chiamata di sistema read e si occupa di richiedere la traduzione dei caratteri letti in interi mediante la funzione \_atoi definita nel file atoi.s salvandoli successivamente nelle rispettive variabili globali. In caso i caratteri letti siano pari a 0 salta all'etichetta \_eof;
- · \_eof: In caso di EOF mette -1 in ebx e ritorna.

## 7 atoi.s

Contiene la funzione che si occupa di convertire una serie di caratteri ASCII in un numero intero.

#### 7.1 Funzioni ed Etichette

- · \_atoi: Vengono inizializzati i registri necessari alla conversione;
- · \_atoi\_loop: Loop principale, converte la stringa puntata dal registro edi in un intero salvato e ritornato in eax.

## 8 check.s

Contiene la funzione che si occupa di settare sulla base dei valori di input e dei valori del ciclo precedente i corretti parametri delle variabili alm, mod, numb.

## 8.1 Funzioni ed Etichette

- · \_check: In base ai valori di init e rpm si occupa di saltare all'etichetta corretta.
- · \_fg / \_sg / \_opt: Etichette corrispondenti alle modalità di funzionamento previste dalle specifiche. Si occupano di settare i corretti valori di alm, mod, numb. L'unica modalità che necessita di una gestione particolare è fg in cui bisogna settare l'eventuale allarme.
- · \_reset\_numb: Nel caso in cui sia stata cambiata la modalità di funzionamento oppure il valore della variabile reset sia pari a 1, viene settata la modalità corretta, resettato il conteggio numb e "spento" alm.
- · \_set\_alm: Se il motore è nella modalità fg da più di 15 secondi, viene "acceso" l'allarme portando il valore di alm a 1.
- · \_init\_0: Se il valore di init è pari a 0 tutte le variabili di output vengono poste a 0 dal momento che il motore è spento.
- · \_end\_check: Abbiamo ritenuto opportuno (per evitare di preoccupare troppo il conducente) considerare un numero di secondi di massimo due cifre. Prima di terminare la funzione si controlla che il valore di numb non sia superiore a 99, in caso si salta a \_numb\_overflow che azzera numb.

## Scelte Progettuali

Si è scelto di dividere il progetto in poche funzioni e di tenere tutti i controlli relativi a input/output in un'unica funzione chiamata \_check. Sarebbe stato tranquillamente possibile fare una divisione in funzioni minori per aumentare la leggibilità (e.g. check\_rpm, check\_init, ...), ma è stato ritenuto più pratico, per la semplicità dei controlli, mantenere il codice in un unico file.

## 9 write line.s

Contiene la funzione che si occupa di creare la stringa di output e scriverla sul corrispondente file. Ricordiamo qui la codifica utilizzata delle modalità di funzionamento e la struttura di una singola riga di output:

Spento	0	00
SG	1	01
OPT	2	10
FG	3	11

alm (1 byte)	,	mod (2 byte)	,	numb (2 byte)	\n
--------------	---	--------------	---	---------------	----

## 9.1 Variabili Locali

- · output\_buff: Buffer di dimensione OUTPUT\_BUFF\_LEN che conterrà i caratteri della stringa da scrivere sul file di output.
- · OUTPUT\_BUFF\_LEN: Dimensione del buffer.
- MOD\_XX: Stringhe costanti che identificano la modalità di funzionamento corrispondente in binario.
- · MOD\_LEN: Dimensione delle stringhe di modalità.

## 9.2 Funzioni ed Etichette

- · \_write\_line: Inizializza i registri necessari alla scrittura sulla variabile buffer.
- · \_alm\_X: Aggiunge al buffer il corretto valore di alm.
- · \_print\_mod: Aggiunge al buffer una virgola come separatore e in base al valore di mod salta alla corrispondente etichetta.
- · \_mod\_X: La stringa corrispondente alla modalità X codificata in binario viene messa in eax, in seguito viene aggiunta al buffer nell'etichetta \_end\_print\_mod.
- · \_print\_numb: Si occupa di aggiungere al buffer il valore di numb opportunamente convertito in ASCII e di terminare la stringa con il carattere \n.

· \_numb\_one\_digit: Se il valore di numb è minore di 10 si occupa di aggiungere uno 0 come prima cifra.

## 10 itoa.s

Contiene la funzione per convertire un valore da intero a una corrispondente stringa ASCII.

## 10.1 Funzioni ed Etichette

- $\cdot$   $\_itoa:$  Inizializza i registri necessari alla conversione.
- · \_itoa\_dividi: Si occupa di contare i caratteri necessari per la stringa e di posizionare il carattere \0 alla fine della stringa.
- · \_itoa\_converti: Scrive ogni cifra nella posizione corretta della stringa.

## 11 close files.s

Contiene la funzione per chiudere correttamente un file.

## 12 Codice Sorgente

## 12.1 syscall.inc

```
# file di definizione delle chiamate di sistema linux
                    SYS_EXIT, 1
                   SYS_READ, 3
           . equ
                   SYS_WRITE, 4
           .equ
                    SYS_OPEN, 5
           . equ
           . equ
                   SYS_CLOSE, 6
                   STDIN, 0
9
           . equ
                   STDOUT, 1
10
           . equ
                   STDERR, 2
           . equ
12
                   SYSCALL, 0x80
13
```

### 12.2 main.s

```
# Progetto Assembly 2016
          # File: main.s
          # Autori: Noé Murr, Mirko Morati
3
          # Descrizione: File principale, punto di inizio del programma.
          .include "syscall.inc"
          .section
                     .data
          input_fd:
                      .long 0
                                     # variabile globale che conterrá il file
9
          # descriptor del file di input
10
11
          output_fd: .long 0
                                     # variabile globale che conterrá il file
12
          # descriptor del file di output
13
14
          # Variabili globali per i segnali di input
15
          init: .long 0
16
          reset: .long 0
17
18
          rpm: .long 0
19
          # Variabili globali per i segnali di output
20
                 .long 0
21
          alm:
          numb:
                  .long 0
22
                  .long 0
23
          mod:
24
          # Codice del programma
26
          .section .text
          .globl input_fd
          .globl output_fd
29
          .globl init
30
          .globl reset
```

```
.globl
                   rpm
           .globl
                   alm
33
           .globl numb
34
           .globl mod
35
           .globl _start
36
37
           # Stringa per mostrare l'utilizzo del programma in caso di parametri errati
38
           usage: .asciz "usage: programName inputFilePath outputFilePath\n"
39
                   USAGE_LENGTH, .-usage
40
           .equ
41
42
           _start:
           # Recupero i parametri del main
43
           popl
                  %eax
                                         # Numero parametri
44
45
           # Controllo argomenti, se sbagliati mostro l'utilizzo corretto
46
           cmpl
                   $3, %eax
47
                   _show_usage
           jne
48
49
           pop1
                   %eax
                                         # Nome programma
50
           pop1
                   %eax
                                         # Primo parametro (nome file di input)
51
                                         # Secondo parametro (nome file di output)
           pop1
                   %ebx
52
53
           # NB: non salvo ebp in quanto non ha alcuna utilitá
54
           # nella funzione start che comunque non ritorna
55
56
                   %esp, %ebp
           mov1
57
58
           call
                   _open_files
                                         # Apertura dei file
59
60
           _main_loop:
61
62
63
           call
                   _read_line
                                        # Leggiamo la riga
64
                   $-1, %ebx
                                         # EOF se ebx == -1
           cmp1
65
66
           jе
                   _end
67
                   _check
                                         # Controllo delle variabili
           call
68
69
           call
                   _write_line
                                        # Scrittura delle variabili di output su file
70
71
                                         # Leggi un altra riga finché non é EOF
           jmp
                   _main_loop
72
73
           _end:
74
75
                                       # Chiudi correttamente i file
                   _close_files
76
           call
77
           # sys_exit(0);
78
                   $SYS_EXIT, %eax
           movl
79
                   $0, %ebx
           movl
80
           int
                   $SYSCALL
81
82
83
           _show_usage:
           # esce in caso di errore con codice 1
84
```

```
# sys_write(stdout, usage, USAGE_LENGTH);
                   $SYS_WRITE, %eax
           mov1
86
                   $STDOUT, %ebx
           movl
87
                  $usage, %ecx
88
           mov1
           mov1
                  $USAGE_LENGTH, %edx
89
           int
                  $SYSCALL
90
91
           # sys_exit(1);
92
                   $SYS_EXIT, %eax
           movl
93
                   $1, %ebx
           movl
94
                  $SYSCALL
           int
95
96
```

## 12.3 open\_files.s

```
# Progetto Assembly 2016
          # File: open_files.s
          # Autori: Noé Murr, Mirko Morati
5
          # Descrizione: File contenente la funzione che si occupa di aprire i file
      di input e di
          # output, i file descriptor vengono inseriti in variabili globali.
          # Si suppone che il nome dei due file siano salvati negli indirizzi
      contenuti
          # rispettivamente in %eax (input) ed in %ebx (output).
9
          .include "syscall.inc"
10
12
          .section .text
13
          error_opening_files: .asciz "errore nell' apertura dei file\n"
14
          .equ ERROR_OPENING_LENGTH, .-error_opening_files
15
16
                                           # Dichiaro la funzione globale
          .globl _open_files
17
                  _open_files, @function # Dichiaro l'etichetta come una funzione
          .type
18
19
          _open_files:
20
21
          pushl
                  %ebp
22
          movl
                  %esp, %ebp
23
24
                                           # Pusho l' indirizzo del file di output
                  %ebx
25
          pushl
          # sullo stack
26
27
                  %eax, %ebx
                                           # Sposto l' indirizzo del file che vado
          movl
28
          # ad aprire in %ebx
30
                  $SYS_OPEN, %eax
                                           # Chiamata di sistema open
          mov1
31
                  $0, %ecx
                                           # read-only mode
          movl
32
                  $SYSCALL
                                           # Apro il file
          int
34
```

```
cmpl
                 $0, %eax
                 _error_opening_files
36
          j1
37
                                        # Metto il file descriptor nella
          movl
                 %eax, input_fd
38
          # sua variabile
40
                                         # Riprendo l' indirizzo del nome del file
          popl
                %ebx
41
          # di output che avevo messo sullo stack
42
43
                 $SYS_OPEN, %eax
                                        # Chiamata di sistema open
         mov1
44
                $01101, %ecx
                                        # read and write mode
45
         mov1
         mov1
                $0666, %edx
                                        # flags
46
         int
                $SYSCALL
                                         # Apro il file
47
48
                 $0, %eax
49
          cmpl
          jl
                 _error_opening_files
                                        # Metto il file descriptor nella
         movl
                %eax, output_fd
         # sua variabile
54
                 %ebp, %esp
         mov1
          pop1
                 %ebp
56
                                         # Ritorna al chiamante
         ret
57
58
         _error_opening_files:
59
         # Esce con codice di errore 2
60
         # sys_write(stdout, usage, USAGE_LENGTH);
61
                $SYS_WRITE, %eax
63
         movl
                 $STDERR, %ebx
               $error_opening_files, %ecx
         movl
64
         66
         int
                $SYSCALL
67
         # sys_exit(2);
68
                $SYS_EXIT, %eax
          movl
          mov1
                 $2, %ebx
70
          int
                $SYSCALL
71
```

## 12.4 read line.s

```
12
           .section .text
13
           .globl _read_line
14
           .type _read_line, @function
15
16
           _read_line:
17
           pushl
                   %ebp
18
                   %esp, %ebp
           movl
19
20
           # Lettura riga
21
           # sys_read(input_fd, input_buff, INPUT_BUFF_LEN);
22
                   input_fd, %ebx
23
                   $SYS_READ, %eax
           mov1
24
                   input_buff, %ecx
           leal
25
                    $INPUT_BUFF_LEN, %edx
26
           mov1
           int
                    $SYSCALL
27
28
                   $0, %eax
                                                  # Se eax == 0 EOF
           cmpl
                    _eof
30
           jе
31
           # Estrazione dei valori di init, reset, rpm dal buffer
32
                   input_buff, %edi
           leal
33
                    _atoi
           call
34
                   %eax, init
           mov1
35
36
           incl
                   %edi
                                                  # Salto il carattere ','
37
38
                    _atoi
           call
39
40
           mov1
                   %eax, reset
41
                   %edi
                                                  # Salto il carattere ','
           incl
42
43
           call
                    _atoi
44
                   %eax, rpm
           mov1
45
46
                   %ebp, %esp
           mov1
47
                   %ebp
           popl
48
49
           xorl
                   %ebx, %ebx
                                                  # ebx = 0 permette di proseguire
50
           ret
51
52
           _eof:
53
           # in caso di EOF \%ebx = -1
54
           movl
                 %ebp, %esp
55
                   %ebp
           pop1
57
                   $-1, %ebx
           mov1
58
           ret
59
```

## 12.5 atoi.s

```
# Progetto Assembly 2016
          # File: atoi.s
          # Autori: Alessandro Righi, Noé Murr, Mirko Morati
3
          # Descrizione: Funzione che converte una stringa in intero.
5
6
7
          .section .text
          .globl _atoi
8
          .type _atoi, @function
9
10
11
          # Funzione che converte una stringa di input in numero
          # Prototipo C-style:
12
          # uint32_t atoi(const char *string);
13
          # Parametri di input:
14
          # EDI - Stringa da convertire
15
          # Parametri di output:
16
          # EAX - Valore convertito
17
18
          _atoi:
19
20
          xorl
                  %eax, %eax  # azzero il registro EAX per contenere il risultato
                  %ebx , %ebx # azzero EBX
21
                  $10, %ecx # sposto 10 in ECX che conterrá il valore
          mov1
22
      moltiplicativo
23
          _atoi_loop:
24
                %ebx, %ebx
          xorl
25
          movb
                  (%edi), %bl # sposto un byte dalla stringa in BL
26
          subb
                  $48, %bl  # sottraggo il valore ASCII dello 0 a BL
27
          # per avere un valore intero
28
29
                  $0, %bl
                                # Se il numero é minore di 0
          cmpb
30
                  _atoi_end
                               # allora esco dal ciclo
          jl.
31
                              # Se il numero é maggiore o uguale a 10
                  $10, %bl
32
          cmpb
33
          jge
                  _atoi_end # esco dal ciclo
34
          mul1
                  %ecx
                                # altrimenti moltiplico EAX per 10
35
          # (10 messo precedentemente in ECX)
36
37
                 %ebx, %eax # aggiungo a EAX il valore attuale
          incl
                  %edi
                               # incremento EDI
38
39
40
          jmp
                  _atoi_loop
                              # rieseguo il ciclo
41
          _atoi_end:
42
43
          ret
44
```

### 12.6 check.s

```
# Progetto Assembly 2016
           # File: check.s
           # Autori: Noé Murr, Mirko Morati
3
           # Descrizione: Funzione che controlla le variabili
5
           # init, reset, rpm e setta le variabili alm, mod e numb
6
7
           .section .data
8
9
           .section .text
10
11
           .globl _check
           .type _check, @function
12
13
14
           _check:
           pushl
                   %ebp
15
           mov1
                   %esp, %ebp
16
17
           # Caso init == 0: alm = 0; mod = 0; numb = 0;
18
           cmp1
                   $0, init
19
                   _init_0
20
           jе
21
           # Caso SG: alm = 0; mod = 1; numb = reset == 1 ? 0 : numb + 1;
22
           cmpl
                   $2000, rpm
23
24
           j1
                   _sg
25
           # Caso OPT: alm = 0; mod = 2; numb = reset == 1 ? 0 : numb + 1;
26
                $4000, rpm
27
           cmpl
           jle
                   _opt
28
29
           # Caso FG: alm = numb >= 15? 1 : 0; mod = 3; numb = reset == 1 ? 0 : numb +
30
       1;
           _fg:
31
           # Salviamo la nuova modalita' in %eax e controlliamo reset
32
33
           mov1
                  $3, %eax
           cmp1
                   $1, reset
34
                   _reset_numb
           jе
35
36
37
           # Se la nuova modalitá non é la stessa si resetta il numero di secondi
           cmpl
                $3, mod
38
                   _reset_numb
           jne
39
40
           incl
                  numb
41
           movl
                   %eax, mod
42
43
           # Se il numero di secondi é maggiore o uguale a 15 viene alzata l'allarme
44
                   $15, numb
           cmpl
45
46
           jge
                    _set_alm
47
           jmp
                   _end_check
48
49
50
           _opt:
```

```
$2, %eax
51
            mov1
            cmpl
                      $1, reset
                      _reset_numb
53
            jе
54
                      $2, mod
55
            cmpl
                      _reset_numb
56
            jne
57
                      numb
            incl
58
            mov1
                      %eax, mod
59
60
            jmp
                      _end_check
61
62
63
            _sg:
                      $1, %eax
            movl
64
                      $1, reset
65
            cmpl
                      _reset_numb
            jе
66
67
            cmp1
                      $1, mod
            jne
                       _reset_numb
69
70
            incl
                      numb
71
                      %eax, mod
72
            mov1
73
                      _end_check
            jmp
74
75
            _reset_numb:
76
                      %eax, mod
            mov1
77
                      $0, numb
78
            mov1
                      $0, alm
79
            mov1
80
            jmp
                      _end_check
81
82
            _set_alm:
83
                     $1, alm
            mov1
84
86
                      _end_check
            jmp
87
            _init_0:
89
            mov1
                      $0, alm
90
                      $0, numb
            mov1
91
                      $0, mod
92
            mov1
93
            _end_check:
94
            # Se il numero di secondi supera i 99 allora dobbiamo ricominciare il
96
       conteggio
                      $99, numb
            cmp1
97
                      _numb_overflow
98
            jg
                     %ebp, %esp
            mov1
99
                      %ebp
100
            popl
101
102
            ret
```

```
103
104    _numb_overflow:
105    movl    $0, numb
106    jmp    _end_check
107
```

## 12.7 write line.s

```
# Progetto Assembly 2016
          # File: check.s
3
          # Autori: Noé Murr, Mirko Morati
          # Descrizione: Funzione che scrive una riga alla volta nel file di output
6
          .include "syscall.inc"
          .section .bss
          .equ OUTPUT_BUFF_LEN, 8
10
          output_buff: .space OUTPUT_BUFF_LEN
12
13
          .section .text
          .globl _write_line
14
          .type _write_line, @function
15
          MOD_00: .ascii "00"
                                # motore spento
16
          MOD_01: .ascii "01"
17
                                       # motore sotto giri
          MOD_10: .ascii "10"
                                       # motore in stato ottimale
18
          MOD_11: .ascii "11"
                                       # motore fuori giri
19
                MOD_LEN, 2
          . equ
20
21
22
          _write_line:
          push1 %ebp
23
          movl
                 %esp, %ebp
24
25
                 output_buff, %edi
26
                                      # spostiamo il puntatore
          # del buffer di output in EDI
27
28
          cmpl
                  $1, alm
                                        # se l'allarme é 1 stampiamo 1
          # altrimenti 0 senza chiamare funzioni
30
                  _alm_1
          jе
31
          _alm_0:
33
                 $48, (%edi)
          movl
34
35
          jmp
                  _print_mod
          _alm_1:
37
          movl
                 $49, (%edi)
38
          _print_mod:
40
          movl $44, 1(%edi)
                                       # aggiungiamo la virgola dopo
41
          # il segnale di allarme
42
          addl $2, %edi
                                        # spostiamo un immaginario cursore
          # nella posizione dove stampare la mod
```

```
$1, mod
                                         # controlliamo il valore di mod
46
          # e stampiamo la stringa corretta in base
47
          # alla giusta modalita' di funzionamento
48
                  _{mod_1}
49
                  $2, mod
          cmpl
50
                   _mod_2
          jе
51
                  $3, mod
          cmpl
52
                   _mod_3
          jе
54
55
          _mod_0:
          movl
                  MOD_00, %eax
56
          jmp _end_print_mod
57
58
59
          _mod_1:
          movl MOD_01, %eax
60
          jmp _end_print_mod
61
          _mod_2:
63
          mov1
                 MOD_10, %eax
64
          jmp _end_print_mod
65
66
          _mod_3:
67
          movl MOD_11, %eax
68
69
          _end_print_mod:
70
          movl %eax, (%edi)
                                         # mettiamo la stringa nell' output_buff
71
                  $MOD_LEN, %edi
                                          # spostato il cursore (la posizione di edi)
72
          addl
73
          # nel punto esatto dove scrivere
          mov1
                $44, (%edi)
                                         # aggiungiamo la virgola
74
                  %edi
          incl
                                          # spostiamo il cursore
75
76
                 $10, numb
                                          # controlliamo se il numero di secondi
77
          # é ad una sola cifra, in tal caso
78
          # aggiungiamola cifra 0
79
                  _numb_one_digit
80
81
          _print_numb:
                 numb, %eax
                                         # prepariamo la chiamata per itoa
          movl
83
84
                                          # chiamiamo itoa
          call
                   _itoa
85
86
87
          leal
                  output_buff, %edi
                                          # mettiamo il puntatore di output_buff in edi
88
          addl
                  $7, %edi
                                          # ci aggiungiamo 7 per arrivare
          # alla fine della stringa,
90
          movl
                  $10, (%edi)
                                         # punto nel quale aggiungiamo un \n
91
92
                  $SYS_WRITE, %eax
          mov1
93
          mov1
                  output_fd, %ebx
94
95
          leal
                   output_buff, %ecx
          mov1
                   $OUTPUT_BUFF_LEN, %edx
97
          int
                   $SYSCALL
```

```
99
             movl
                       %ebp, %esp
100
                       %ebp
101
             popl
102
             ret
104
             _numb_one_digit:
105
                       $48, (%edi)
             movl
106
             incl
                       %edi
107
                       _print_numb
108
             jmp
109
```

#### 12.8 itoa.s

```
# Progetto Assembly 2016
          # File: itoa.s
          # Autori: Alessandro Righi, Noé Murr, Mirko Morati
3
          # Descrizione: Funzione che converte un intero in stringa
6
          # Prototipo C-style:
             u_int32_t itoa(uint32_t val, char *string);
          # Parametri di input:
            EAX - Valore intero unsigned a 64bit da convertire
9
            EDI - Puntatore alla stringa su cui salvare il risultato
10
          # Parametri di output:
              EAX - Lunghezza della stringa convertita (compresiva di \0 finale)
12
          .section .text
14
          .global _itoa
          .type _itoa, @function
16
17
          _itoa:
18
19
          movl
                  $10, %ecx
                              # porto il fattore moltiplicativo in ECX
                  %eax, %ebx # salvo temporaneamente il valore di EAX in EBX
          movl
20
                  %esi, %esi # azzero il registro ESI
          xorl
21
22
          _itoa_dividi:
23
                  %edx , %edx # azzero EDX per fare la divisione
          xorl
24
          divl
                  %ecx
                               # divide EAX per ECX, salva il resto in EDX
25
          {\tt incl}
                               # incrementa il contatore
26
                  %eax, %eax # se il valore di EAX non é zero ripeti il ciclo
          testl
27
28
          jnz
                  _itoa_dividi
29
                  %esi, %edi # somma all'indirizzo del buffer
30
          # il numero di caratteri del numero
31
                  %ebx, %eax # rimette il valore da convertire in EAX
          mov1
                  %esi, %ebx # salvo il valore della lunghezza della stringa in EBX
33
          mov1
34
                  $0, (%edi) # aggiungo un null terminator alla fine della stringa
          movl
35
          decl
                  %edi
                               # decremento il contatore della stringa di 1
37
```

```
_itoa_converti:
                  %edx , %edx # azzero EDX per fare la divisione
39
          xorl
                  %ecx  # divido EAX per ECX, salvo il valore del resto in EDX
          divl
40
                  $48, %edx # sommo 48 a EDX
          addl
41
                  %dl, (%edi) # sposto il byte inferiore di EDX (DL)
42
          # nella locazione di memoria puntata da EDI
43
                  %edi
                              # decremento il puntatore della stringa
          decl
44
                              # decremento il contatore
45
          decl
                  %esi
          testl
                  %esi, %esi # se il contatore non é 0 continua ad eseguire il loop
46
                  _itoa_converti
47
          jnz
48
                  %ebx, %eax # porto il valore della lunghezza
          movl
49
          # della stringa in EAX per ritornarlo
50
                           # incremento di 1 EAX (in modo da includere il \0)
                  %eax
51
          incl
52
          ret
53
```

## 12.9 close files.s

```
# Progetto Assembly 2016
          # File: check.s
          # Autori: Noé Murr, Mirko Morati
3
5
          # Descrizione: Funzione che chiude i file aperti precedentemente
6
          .include "syscall.inc"
8
          .section .text
9
                                            # Dichiaro la funzione globale
          .globl _close_files
10
          .type _close_files, @function # Dichiaro l' etichetta come una funzione
12
          _close_files:
14
15
          pushl
                   %ebp
          movl
                   %esp, %ebp
17
          # sys_close(input_fd);
                 $SYS_CLOSE, %eax
19
          mov1
                   input_fd, %ebx
20
                   $SYSCALL
21
          int
22
          # sys_close(output_fd);
                $SYS_CLOSE, %eax
24
          movl
          movl
                   output_fd, %ebx
25
                   $SYSCALL
          int
26
27
          mov1
                   %ebp, %esp
29
          pop1
                   %ebp
30
31
          ret
```